
78 Wilmott magazine

From the point of view of the sell side, aiming to minimize market
risk by delta hedging, their main exposure is to volatility risk. However,
the contract is very subtle in its dependence on the assumed model 
for volatility.

In this brief note, I will show how the contract value depends on the
treatment of volatility. In particular, I shall show results for constant

volatility and volatility ranges.

2 The subtle nature of the 
cliquet option
Traditionally one measures sensitivity to volatility via
the vega. This is defined as the derivative of the
option value with respect to a (usually constant)
volatility. This number is then used to determine how
accurate a price might be should volatility change. As
part of one’s risk management, perhaps one will vega
hedge to reduce such sensitivity.

This is entirely reasonable when the contract in
question is an exchange-traded vanilla contract and
one is measuring sensitivity to the market’s (implied)
volatility.

However, when it comes to the risk management of
exotic options the sensitivity to a constant volatility is
at best irrelevant and at worst totally misleading. By
now this is common knowledge and I don’t need to

1 Introduction
Cliquet options are at present the height of fashion in the world of equity
derivatives. These contracts, illustrated by the term sheet below, are appeal-
ing to the investor because of their protection against downside risk, yet
with significant upside potential. Capping the maximum, as in this global-
ly floored, locally capped example, ensures that the payoff is never too
extreme and therefore that the value of the contract is not too outrageous.

dwell on the details. It suffices to say that whenever a contract has a Gamma
which changes sign (as does any ‘interesting’ exotic) vega may be small at
precisely those places where sensitivity to actual volatility is very large.

Confused? As a rule of thumb if you increase volatility when Gamma
is positive you will increase a contract’s value. At points of inflection in
the option value (where Gamma is zero) the option value may hardly
move. But this is sensitivity to a parameter that takes the same value
everywhere. What if you increase volatility when Gamma is positive and
decrease it when Gamma is negative? The net effect is an increase in
option value even at points of inflection.

Skews and smiles can make matters even worse, unless you are fortu-
nate enough that your skew/smile model forecasts actual volatility behav-
ior accurately.

The classical references to this phenomenon are Avellaneda, Levy &
Para‘s (1995) and Lyons (1995) but also see Wilmott (2000).

And the relevance to cliquet options? To see this you just need to plot
the formula

max

(
0, min

(
Cap,

Si − Si −1

Si −1

))

against S to see the non convex nature of the option price; Gamma
changes sign.

Now comes the subtle part. The point at which Gamma changes sign
depends on the relative move in S from one fixing to the next. The point of inflec-
tion is not near any particular value of S. The conclusion has to be that
any deterministic, volatility surface model fitted to vanilla prices is not
going to be able to model the risk associated with changing volatility.
This is true even if you allow the local volatility surface to move up and
down and to rotate.

For this reason we are going to focus on using the uncertain volatility
model described in the above-mentioned references. In this model the
actual volatility is chosen to vary with the variables in such a way as to give
the option value its worst (or best) possible value. The actual volatility is
assumed to lie in the range σ − to σ + . The worst option value is when actu-
al volatility is highest for negative Gamma and lowest for positive Gamma:

σ (�) =
{

σ + if � < 0
σ − if � > 0.

Now let us look at the pricing of the cliquet option.

3 Path dependency, constant volatility
We will be working in the classical lognormal framework for the
underlying

dS = µS dt + σ S dX.

Assuming for the moment that volatility is constant, or at most a
deterministic function of stock price S and time t, we can approach the
pricing from the two most common directions, Monte Carlo simulation
and partial differential equations. A brief glance at the term sheet shows

that there are none of the nasties such as early exercise, convertibility or
other decision processes that make Monte Carlo difficult to implement.

3.1 Monte Carlo
Monte Carlo pricing requires a simulation of the risk-neutral random
walk for S, the calculation of the payoff for many, tens of thousands,
say, of paths, and the present valuing of the resulting average payoff.
This can be speeded up by many of the now common techniques.
Calculation of the greeks is slightly more time consuming but still
straightforward.

3.2 PDE
To derive a partial differential equation which one then solves via, for
example, finite-difference methods, requires one to work out the
amount of path dependency in the option and to count the number of
dimensions. This is not difficult, see Wilmott (2000).

In all non trivial problems we always have the two given dimensions, S
and t. In order to be able to keep track, before expiry, of the progress of the
possible option payoff we also need the following two new ‘state variables’

S ′ and Q ,

where

S ′ = the value of S at the previous fixing = Si

and
Q = the sum to date of the bit inside the max function

=
i∑

j=1

max

(
0, min

(
Cap,

S j − S j−1

S j−1

))
.

Here I am using the index i to denote the fixing just prior to the current
time, t. This is all made clear in the figure.

Since S ′ and Q are only updated discretely, at each fixing date, the
pricing problem for V(S, t, S ′, Q ) becomes

∂V

∂ t
+ 1

2 σ 2S 2 ∂2V

∂S 2
+ rS

∂V

∂S
− rV = 0

where r is the risk-free interest rate. In other words, the vanilla
Black–Scholes equation. The twist is that V is a function of four variables,
and must further satisfy the jump condition across the fixing date

V(S, t −
i , S ′, Q ) = V

(
S, t +

i , S, Q + max

(
0, min

(
E1,

S − S ′

S ′

)))

and the final condition

V (S, T, S ′, Q ) = max(Q , E2).

Here E1 is the local cap and E2 the global floor. (More general payoff
structures can readily be imagined.)

Being a four-dimensional problem it is a toss up as to whether a
Monte Carlo or a finite-difference solution is going to be the faster.

However, the structure of the payoff, and the assumption of lognormali-
ty, mean that a similarity reduction is possible, taking the problem
down to only three dimensions and thus comfortably within the domain
of usefulness of finite-difference methods. The similarity variable is

ξ = S

S ′ .

The option value is now a function of ξ , t and Q. The governing equation
for V (ξ, t, Q ) (loose notation, but the most clear) is

∂V

∂ t
+ 1

2 σ 2ξ 2 ∂2V

∂ξ 2
+ r ξ

∂V

∂ξ
− rV = 0.

The jump condition becomes

V(ξ, t −
i , Q ) = V

(
1, t +

i , Q + max (0, min (E1, ξ − 1))
)

and the final condition is

V(ξ, T, Q ) = max(Q , E2).

All of the results that I present are based on the finite-difference solu-
tion of the partial differential equation. The reason for this is that I want
to focus on the volatility dependence, in particular I need to be able to
implement the uncertain volatility model described above and this is not
so simple to do in the Monte Carlo framework. (The reason being that
volatility depends on Gamma in this model and Gamma is not calculated
in the standard Monte Carlo implementation.)

4 Results
The following results are based on the cliquet option described in the
term sheet. In particular, it is a five-year contract with annual fixings, a
global floor of 16% and local caps of 8%. The interest rate is 3% and there
are no dividends on the underlying.

To understand the following you must remember that the cliquet
value is a function of three independent variables, ξ, Q and t. I will be
showing plots of value against various variables at certain times before
expiry. These will assume a constant volatility. Then we will look at the
effect of varying volatility on the prices.

4.1 Constant volatility
In the following five plots volatility is everywhere 25%.
This plot shows the cliquet value against Q and ξ at 4.5 years before expiry.
The contract has thus been in existence for six months. At this stage there
have been no fixings yet and
the state variable Q only
takes the value 0. The non
convex contract value can
be clearly seen.

At 3.5 years before
expiry, and therefore 1.5
years into the contract’s
life, the value is shown in
the next figure, above. The
state variable Q now ranges
from zero to 8%.

One year later, see the
next figure above, the con-
tract is exactly half way
through its life. The state
variable Q lies in the range
zero to 16%. For small val-
ues of Q the option value is
very close to being the
present value of the 16%
floor. This represents the
small probability of get-
ting a payoff in excess of
the floor at expiry.

After being in exis-
tence for 3.5 years, and
having only 1.5 years left
to run, the cliquet value is
as above. Now Q ranges
from zero to 24%. When Q
is zero there is no chance
of the global floor being

exceeded and so the contract value there is exactly the present value of
16%.

Six months before expiry the option value is as shown above. Q ranges
from zero to 32% and for any values below 8% the contract is again only
worth the present value of 16%.

4.2 Uncertain volatility
The above shows the evolution of the option value for constant volatility.
There is diffusion in the ξ direction and a ‘jump condition’ to be applied
at every fixing. The amount of the diffusion is constant. (Or rather, is con-
stant on a logarithmic scale.)

To price the contract when volatility is uncertain we must use a volatil-
ity that depends on (the sign of) Gamma. Some results are shown below.

In the next figure is plotted the contract value against ξ at five years
before expiry with Q = 0. Five calculations have been performed.

1. The first line to examine is the fuschia-colored line in the fig-
ure, the middle line. This corresponds to a constant volatility
of 25%. This is the base case with which we compare other
prices.

2. The second line to examine is the blue one, close to the middle
line. This is the cliquet value with a constant volatility of just
20%.

3. The third case has a constant volatility of 30%. This is the orange
line in the figure.

4. The fourth line is green, representing the cliquet value when
the volatility is allowed to range between 20 and 30%, taking a
value locally that maximizes the cliquet value overall 

5. The fifth and final curve is the purple one for which volatility
has again been allowed to range from 20 to 30% but now such
that it gives the option its lowest possible value.

The first observation to make is how close the constant volatility curves
are, i.e. curves 1–3. As stated above, a good rule of thumb is that high volatil-
ity and positive Gamma give a high option value. Because Gamma changes
sign in this contract a result of this is that there is a ξ value at which the
contract value does not appear to be sensitive to the volatility. In this case
the value is around 0.95, close to the point of inflection.

Now ask yourself the following question. “Do I believe that volatility
is a constant, and this constant is somewhere between 20% and 30%? Or
do I believe that volatility is highly uncertain, but is most likely to stay
within the range 20% to 30%?”

If you believe the former, then the calculation we have just done, in
curves 1–3, is relevant. If, on the other hand, you think that the latter is
more likely (and who wouldn’t?) then you must discard the calculations
in curves 1–3 and consider the whole spectrum of possible option values
by looking at the best and worst cases, curves 4 and 5.

Such calculations show that the real sensitivity to volatility is much,
much larger than a naive vega calculation would suggest.

The next table shows how the cliquet value (five years before expiry at
Q = 0 and ξ = 1) varies with the allowed range for volatility. The table is
to be read as follows. When volatility takes one value only, read along the
diagonal, the orange cells, to see the contract values. For example, when
the volatility is 22% the contract value is 0.1739. And when the volatility
is 27% the contract value is 0.1726.

Now consider a range of possible volatilities. Suppose you believe
volatility will not stray from the range 22% to 27%. The worst case is is in
the yellow cells, in this case 0.1647. The best case is to be found in the
clear cells, 0.1830. So, when volatility ranges from 22 to 27% the correct
range for the contract value is 0.1647 to 0.1830.

When volatility is a constant, but a constant between 22% and 27%,
the contract value range is 0.1739 − 0.1726 = 0.0013 or 0.75% relative (to
mid price) range. When volatility is allowed to varying over the 22–27%
range we find that the contract value itself has a value range of
0.1830 − 0.1647 = 0.0183 or 10.5% relative (to mid price) range. The true
sensitivity to volatility is 14 times greater than that estimated by vega.

5 Code sample: Cliquet with uncertain
volatility, in similarity variables
Below is some Visual Basic code that can be used for pricing these cliquet
options in the uncertain volatility framework.

The range for volatility is  VolMin to VolMax, the dividend yield is
Div, risk-free interest rate IntRate, the local cap is Strike2 and the
global floor Strike1. Expiry is Expiry. The numerical parameter is
NumAssetSteps, the number of steps in the S and Q directions.

This program clearly leaves much to be desired, for example in the
discretization, the treatment of the jump condition etc. But it does have
the benefit of transparency.

A: The timestep is set so that the explicit finite difference method is
stable. If the timestep is any smaller than this the method will not con-
verge.

B: Here the payoff is set up, the dependent variable as a function of the
independent variables.

C: The timestepping engine. Delta and Gamma are discretized versions
of the first- and second-order derivatives with respect to S. This part of
the code also treats the uncertain volatility. See how the volatility
depends on the sign of Gamma.

D: The boundary conditions, for ξ = 0 and large ξ .

E: Updating the next step back in the grid.

F: Here the code tests for a fixing date.

G: Across fixing dates the updating rule is applied. This is really the only
point in the code that knows we are pricing a cliquet option.

the Lagrangian volatility model. Applied Mathematical Finance 3 21–53

Vmax
0.2 0.21 0.22 0.23 0.24 0.25 0.26 0.27 0.28 0.29 0.3

0.2 0.1743 0.1720 0.1700 0.1680 0.1662 0.1645 0.1629 0.1615 0.1601 0.1588 0.1576
0.21 0.1763 0.1741 0.1719 0.1699 0.1680 0.1663 0.1646 0.1631 0.1617 0.1603 0.1591
0.22 0.1784 0.1761 0.1739 0.1718 0.1698 0.1680 0.1663 0.1647 0.1632 0.1618 0.1605
0.23 0.1804 0.1780 0.1757 0.1736 0.1716 0.1698 0.1680 0.1663 0.1648 0.1633 0.1620
0.24 0.1824 0.1799 0.1776 0.1754 0.1734 0.1715 0.1696 0.1679 0.1663 0.1648 0.1634
0.25 0.1843 0.1818 0.1794 0.1772 0.1751 0.1731 0.1713 0.1695 0.1679 0.1663 0.1648
0.26 0.1863 0.1837 0.1812 0.1789 0.1768 0.1748 0.1729 0.1711 0.1694 0.1678 0.1662
0.27 0.1881 0.1855 0.1830 0.1807 0.1785 0.1764 0.1744 0.1726 0.1708 0.1692 0.1676
0.28 0.1900 0.1873 0.1847 0.1824 0.1801 0.1780 0.1760 0.1741 0.1723 0.1706 0.1690
0.29 0.1918 0.1890 0.1865 0.1840 0.1817 0.1796 0.1775 0.1756 0.1738 0.1720 0.1704
0.3 0.1935 0.1908 0.1881 0.1857 0.1833 0.1811 0.1790 0.1770 0.1752 0.1734 0.1717

m

Option Explicit 
Function cliquet(VolMin, VolMax, Div, IntRate, Strike1, Strike2, NumFixes, Fixing, Expiry, NumAssetSteps) 
ReDim xi(-NumAssetSteps To NumAssetSteps) 
Dim Vmax, AssetStep, TStep, QStep, Delta, Gamma, Theta, Tim, Vol, qafter, frac, V1, V2 As Double 
Dim i, j, k, M, iafter, kafter, N, NumSoFar, NumQSteps As Integer 
ReDim jtest(1 To NumFixes) As Integer 
 
Vmax = Application.Max(VolMin, VolMax) 
AssetStep = 1 / NumAssetSteps 
TStep = 0.95 * AssetStep ^ 2 / Vmax ^ 2 / 2 ^ 2 ' This ensures stability of the explicit method 
M = Int(Expiry / TStep) + 1 
TStep = Expiry / M 
QStep = AssetStep 
NumQSteps = Int(Strike2 / QStep) * NumFixes 
 
ReDim Q(0 To NumQSteps) 
ReDim VOld(-NumAssetSteps To NumAssetSteps, 0 To NumQSteps) ' First dimension centred on xi = 1 
ReDim VNew(-NumAssetSteps To NumAssetSteps, 0 To NumQSteps) 
 
NumSoFar = 1 
For j = 1 To NumFixes - 1 
jtest(j) = Int(j * Fixing / TStep) ' Used in testing whether fixing date has been passed 
Next j 
 
For k = 0 To NumQSteps 
Q(k) = k * QStep 
For i = -NumAssetSteps To NumAssetSteps 
xi(i) = 1 + AssetStep * i ' i = 0 corresponds to xi = 1 
VOld(i, k) = Application.Max(Strike1, Q(k) + _ 
                    Application.Max(0, Application.Min(Strike2, xi(i) - 1))) ' Payoff 
Next i 
Next k 
 
For j = 1 To M 
 
For k = 0 To NumQSteps 
For i = -NumAssetSteps + 1 To NumAssetSteps - 1 
Delta = (VOld(i + 1, k) - VOld(i - 1, k)) / 2 / AssetStep ' Central difference 
Gamma = (VOld(i + 1, k) - 2 * VOld(i, k) + VOld(i - 1, k)) / AssetStep / AssetStep 
Vol = VolMax 
If Gamma > 0 Then Vol = VolMin ' Volatility depends on Gamma in the uncertain volatility model 
Theta = IntRate * VOld(i, k) - 0.5 * Vol * Vol * xi(i) * xi(i) * Gamma _ 
                            - (IntRate - Div) * xi(i) * Delta ' The Black-Scholes equation 
VNew(i, k) = VOld(i, k) - TStep * Theta 
Next i 
 
VNew(-NumAssetSteps, k) = VOld(-NumAssetSteps, k) * (1 - IntRate * TStep) ' Boundary condition at xi = 0 
VNew(NumAssetSteps, k) = VNew(NumAssetSteps - 1, k) ' Boundary condition at xi = infinity. Delta = 0 
 
For i = -NumAssetSteps To NumAssetSteps 
VOld(i, k) = VNew(i, k) 
Next i 
 
Next k 
 
If jtest(NumSoFar) = j Then ' Test for a fixing date 
 
    For i = -NumAssetSteps To NumAssetSteps 
    For k = 0 To NumQSteps 
    qafter = Q(k) + Application.Max(0, Application.Min(Strike2, xi(i) - 1)) ' The updating rule 
    kafter = Int(qafter / QStep) 
    frac = (qafter - QStep * kafter) / QStep 
     
    V1 = 0 
    V2 = 0 
    If kafter < NumQSteps Then 
    V1 = VNew(0, kafter) 
    V2 = VNew(0, kafter + 1) 
    End If 
     
    VOld(i, k) = (1 - frac) * V1 + frac * V2 ' The jump condition. Linear interpolation 
    Next k 
    Next i 
    NumSoFar = NumSoFar + 1 
 
End If 
 
Next j 
 
cliquet = VOld ' Output the whole array 
 
End Function

